Simulate MDI without Multiple Document Interface overhead 

by Brendan Delumpa 

Because of space considerations, the full source listings for this article don’t appear in the journal. You can use anonymous FTP to download a ZIP file of the entire project from ftp.cobb.com/ddj. The filename is SIM_MDI.ZIP. 

The Multiple Document Interface 

(MDI) has been a mainstay GUI application paradigm for Windows since its inception. Though Microsoft now dismisses it as an outdated interface standard to be supplanted by the Single Document Interface (SDI), it seems that many software developers just can’t get away from it. And to this day, you’ll see that many programs are still built around MDI, despite Microsoft’s recommendation. 

MDI remains popular among developers - as evidenced by the software packages that even today follow MDI conventions. I’d say this popularity is due to the natural application integration MDI affords developers. Think about it for a moment: Every "child" window is contained in a single, main application window, so it isn’t necessary to think about screen locality or to perform really complex window management routines. Windows offers built-in support that makes MDI window management easy and convenient. And since MDI has been around for several years, Microsoft has had time to refine it. 

These features all contribute to a very attractive foundation on which to build applications. It’s no wonder that MDI applications remain so popular. Unfortunately, all the ease and convenience comes with a lot of resource baggage, not to mention some oddities that many developers have run across at one time or another. Let’s take a look at some of these quirks. 

Overcoming MDI’s shortcomings 

By default, programs create MDI child windows when they start, and those windows remain in memory until the program ends. This retention isn’t such a bad thing if the child windows are fairly simple forms with just a few components. However, if an MDI program has several child windows containing multiple embedded components, then you may run into resource problems. One way to avoid such problems is to create the children only when you need them and destroy them as soon as you’ve finished with them. That is, never create the child windows at program startup. 

MDI forms frequently don’t act like non-MDI forms in some very important respects. For instance, if you set the FormStyle property of an MDI child window to wsMaximized, the child window first pops up in a normal state and then maximizes, producing an annoying flash. One way to fix this behavior is to enclose the window creation of the child window within a LockWindowUpdate block. The following code illustrates how you accomplish this enclosure: 

procedure TForm1.Button1Click(Sender: TObject);

begin

  LockWindowUpdate(Handle);

  Form2 := TForm2.Create(Application);

  Form2.Show;

  LockWindowUpdate(0);

end;

The LockWindowUpdate() function delays screen updates for a window specified by its handle; it doesn’t refresh the screen until you call LockWindowUpdate(0). Any code enclosed within this "block" will run without affecting anything onscreen. This is brute-force programming at its best. 

Furthermore, unlike non-MDI forms, MDI child forms in Delphi 1.0 can’t be hidden. According to Microsoft, hiding an MDI form creates some internal problems in Windows 3.1 itself. Thus the Windows API, by default, prevents you from hiding forms. The solution (which I won’t go into here) involves making some Windows API calls to hide the windows - a dicey proposition at best, and not really recommended. You’re probably better off if you simply destroy the window using 

Action := caFree;

in the OnClose event handler to remove the form from view. Remember, you can always recreate the form if you need to. 

All in all, while these examples don’t add up to much, they do indicate the existence of some internal complexities and quirks in the MDI specification. You may want to avoid these problems, especially if what you’re intending to write doesn’t require all of MDI’s functionality. You can avoid some problems by performing the Windows trickery that I’ll demonstrate below. 

From a form to a child window 

Changing a form into a child window can be as easy as specifying a parent in the form’s OnCreate handler. Look at the example below: 

procedure TForm2.OnCreate(Sender : TObject);

begin

  Parent := Application.MainForm;

end;

There’s absolutely nothing wrong with following this example. In fact, it works quite well though it isn’t very elegant. A better place to specify parentage would be in the form’s virtual method, called CreateParams. This method initializes a TCreateParams structure that is passed to the Windows API function CreateWindowEx, which is responsible for creating all windows in Windows. 

The TCreateParams structure passes vital initialization information to CreateWindowEx, so that the latter knows how to display a window. Delphi automatically handles parameters that you set through the Object Inspector and provides default values. But it is possible to override the CreateParams() function and change the parameters to suit your particular initialization requirements. Take a look at the following structure: 

type

  TCreateParams = record

    Caption: PChar;

    Style: Longint;

    ExStyle: Longint;

    X, Y: Integer;

    Width, Height: Integer;

    WndParent: HWND;

    Param: Pointer

    WindowClass: TWndClass;

    WinClassName: array[0..63] of Char;

  end;

As you can see, there’s a lot you can change here. For our discussion, though, we’re interested in the Style field, which controls the physical display of the form. While its type is LongInt, it’s actually a bit field that stores constant values that you can set using bit-wise operations. 

To make a form a child, you append OR and the window-style constant WS_CHILD to the bit field, like so: 

Style :=  (Style OR WS_CHILD);

However, the Windows help file states that you can’t include the window style WS_POPUP in the bit field if a window is designated as a child. So in addition to applying the WS_CHILD style, we have to remove the WS_POPUP value. 

Therefore, it’s actually better to write the following: 

Style :=  (Style OR WS_CHILD) AND (NOT WS_POPUP);

Unfortunately, this bit of rewriting isn’t all we have to do here. We must also notify both Windows and Delphi of the impending relationship - just to cover our bases. 

We’ll do this by assigning the WndParent field to the parent of our choice. For pseudo-MDI applications, the parent will be the main form, as in 

WndParent := Application.MainForm.Handle;

Finally, to let Delphi know about the relationship, we set the Parent property of the form to the main form of our application: 

Parent := Application.MainForm;

The complete CreateParams method looks like this: 

procedure TChildFrm.CreateParams(var Params : TCreateParams);

begin

  inherited CreateParams(Params);

  with Params do begin

    {Set the child window's style - WS_CHILD

     is absolutely required but you can 

     optionally set other flags as well.}

    Style := (Style OR WS_CHILD) AND (NOT WS_POPUP);

    {Designate the child form's parent in 

     Windows}

    WndParent := Application.MainForm.Handle;

  end;

  {Let Delphi know about the relationship 

   now}

  Parent := Application.MainForm;

end;

I realize that implementing this code is more work than just setting the Parent property to the Application.MainForm in the OnCreate event handler. But what we’re actually doing is working in Delphi behind the scenes and at a much lower level to change the form’s behavior. 

In fact, since you declared the CreateParams() method as a protected virtual method, you can put this form in the object repository for easy subclassing. Granted, you could achieve the same effect by applying the above technique to existing forms, but that’s really just setting a property at run-time and not truly adjusting the form‘s behavior. Here, we’re using the object-oriented programming (OOP) concept of inheritance to create a true descendant of TForm. In any case, we’re still not done; we have one more item on the agenda. 

Window topping 

A couple of phenomena occur when you make a form act as a non-MDI child of another. First of all, the form loses its ability to be a top-level window. What this loss means is that even though you can manipulate it like a regular form, the form’s caption bar will always be colored with the inactive form color (typically gray). No matter what you do, you’ll never get true focus or activation of the form because it’s no longer a top-level window. 

MDI children don’t have this problem because the MDI specification handles the Z-order of child forms. Figure A illustrates the phenomenon of the non-MDI child window. 

Figure A:

PRIVATE "TYPE=PICT;ALT=Figure A"
If you click on Form3 in Figure A, you’d expect its caption bar color to change to blue (or whatever active form color you have set in your system). You can move the form, resize it, and manipulate it to your heart’s content, but it will never activate. 

Furthermore, let’s say you create and display two child forms inside the main window. Because you can’t activate either form, clicking on a child that’s underneath another child won’t move that form to the top of the Z-order (the order of the windows from front to back). Figure B illustrates this problem. 

Figure B:

As you can see in Figure B, two child forms appear on the main form, but both look like inactive forms. The forms’ appearance can be a problem because it may confuse users as to which is the active form. 

Now, to get around these snags, you might think of setting the ExStyle parameter to WS_EX_TOPMOST in the TCreateParams structure to force the form to be a top-level window. But these settings won’t work. In fact, you’ll get an access violation error and crash your program because top-levelness is a Windows trait, not a Delphi trait; therefore, you can’t manipulate the setting without causing problems. So now what? You trick Windows. 

Tricking Windows 

The trickery involved here is nothing complex. On the contrary, it’s merely a matter of sending an activation message to the child form under the proper circumstances. 

When Windows detects the activation message, it changes the color of the caption bar and "activates" the form. The specific message we send is WM_NCACTIVATE, which is the Windows message sent to a window when it’s time to change a window’s non-client area (caption and border) to indicate an active or inactive state. 

You might be wondering why we don’t send the WM_ACTIVATE message instead of WM_NCACTIVATE. We don’t send this particular message because it works only on top-level windows. Since our child form isn’t a top-level window, nothing would happen if we sent a WM_ACTIVATE message to it. On the other hand, the WM_NCACTIVATE message works on any window, so it seems natural to use this message. 

A form can receive a WM_NCACTIVATE message in three distinct circumstances: when the application creates it at runtime, when the user clicks (with either the right or left mouse button) on the window’s client area, and when the user clicks on the window’s caption bar (again, using either the left or right mouse button). There’s a corresponding Windows message for each of these situations, and you can trap and handle them separately. 

With Delphi, it’s a matter of writing a custom message handler. You’ve probably seen Delphi message handlers before; they’re usually in the following form: 

procedure

WM<WinMessage>(var Msg: T<WinMessage>);

message WM_<WinMessage>;

For example, to override the message handler for a left-click, your declaration would look like the following: 

procedure

WMLButtonDown(var Msg : TWMLButtonDown);

message WM_LBUTTONDOWN;

You could actually name the handler anything you prefer, but convention dictates that you name it the closest approximation of the Windows message you’re handling. 

I mentioned earlier that you can activate a form in three circumstances (actually, in five, if you count the mouse-clicks separately) and that each circumstance has an associated Windows message. Therefore, it’s natural to assume that we’d have to write five custom message handlers for each of the messages. In many cases, that’s correct, but since we want to do the same for each circumstance, writing that many procedures is a bit of a waste. 

In this case, we need to turn to a procedure that intercepts all the messages that go to a window. Doing so enables us to handle several messages in one fell swoop. This particular method is another virtual method of TForm called WndProc. 

The WndProc method receives all messages sent to a form and dispatches them accordingly. It’s the ideal place to intercept messages that trigger identical actions. 

For our purposes, we’re interested in the following messages: WM_CREATE, which is sent to WndProc when a form is created; WM_LBUTTONDOWN and WM_RBUTTONDOWN, which are sent to WndProc when a user presses the left or right mouse button, respectively, in the client area of a form; and finally, WM_NCLBUTTONDOWN and WM_NCRBUTTONDOWN, which are sent in response to left or right mouse clicks in a form’s caption or border area. We’d handle these messages in WndProc with the following lines of code: 

procedure TChildFrm.WndProc(var Msg : TMessage);

begin

{Call the default Window procedure so all default actions are taken on all messages}

inherited WndProc(Msg);

{If the following occur, take further action}

if (Msg.Msg = WM_CREATE) OR

{window create}

(Msg.Msg = WM_LBUTTONDOWN) OR

{left-click in client area}

(Msg.Msg = WM_RBUTTONDOWN) OR

{right-click in client area}

(Msg.Msg = WM_NCLBUTTONDOWN) OR

{left-click in non-client area}

(Msg.Msg = WM_NCRBUTTONDOWN)

then {right-click in non-client area}

begin

{"Trick" Windows into thinking this is an active, top-level form.}

Perform(WM_NCACTIVATE, Ord(True), 0);

{Change all other forms of the same class to look inactive.}

ActivateForm;

end

end;

You’ll want the method to first call the inherited WndProc procedure to ensure that all messages get dispatched. (If you fail to do this, you’ll get some really abnormal behavior.) Then, if the messages in which we’re interested do occur, we do more processing. In this particular case, we dispatch another method to the form - the  WM_NCACTIVATE message! 

Notice that we use WM_NCACTIVATE with the Perform procedure. This procedure allows a control to send a message to itself, completely bypassing the Windows message queue and communicating directly with the control’s window procedure (WndProc). Now you can see why calling the inherited WndProc was vital: By sending WM_NCACTIVATE to the form, we ensure that the overridden WndProc processes all messages. So our activation message gets processed. In fact, this one message makes the child forms behave in the way you’d expect MDI child forms to behave relative to their main forms! Unfortunately, we’re not finished. 

Since the child forms aren’t really activated, Windows essentially ignores them. Therefore, once you make the change to the caption bar, the caption bar retains the change. Similarly, once a child form is "activated" by one of the above messages, it retains its look of activation. 

The results in this case are similar to those in Figure B, only now, all of the caption bars are the active form color. We’re in no better shape than we were before. But don’t lose hope - there’s a solution. 

You may have noticed that our code makes a call to a method called ActivateForm. This custom method more accurately acts as a deactivator of other child forms that are of the same class as the one most currently activated, but are not the current form. Here’s the code for that method: 

procedure TChildFrm.ActivateForm;

var

I : Integer;

begin

inherited;

{Set the private ChildHndl variable to

Self.Handle, so we know which form is

active}

ChildHndl := Self.Handle;

Self.BringToFront;

with Screen do begin

{Go through all the forms on screen.}

for I := 0 to FormCount - 1 do

if (Forms[I].ClassName =

Self.ClassName) then

if (Forms[I].Handle <> ChildHndl)

then

{If a "Child" is of the same

type, but not the current one,

as specified by ChildHndl then

make it inactive}

SendMessage(Forms[I].Handle,

WM_NCACTIVATE, Ord(False), 0);

end;

end;

As you can see in the code, the procedure sets a THandle variable called ChildHndl (which is defined in the private sectionof the form) to the value of the current form’s handle. Then it iterates through all the forms on the screen. If a form has the same ClassName value as the current form but has a different Handle value than ChildHndl, a WM_NCACTIVATE with a wParam of False is sent to the window to "deactivate" it. The net result appears in Figure C. 

Figure C:

PRIVATE "TYPE=PICT;ALT=Figure C"Does this result remind you of something? That’s right - it looks just like an MDI application! And at this point, we’re finally finished. 

Conclusion 

Whew! We’ve just covered a lot of ground! I suggest that you look deeper into CreateParams and WndProc because this article demonstrates only the tip of the iceberg of what you can accomplish with these very useful methods. Besides, getting to know those procedures in greater depth will help you gain an even better understanding of the inner workings of your Delphi programs. 

